Math 173
Calculus I

Dr. Constant J. Goutziers
Department of Mathematics, Computer Science and Statistics
goutzicj@oneonta.edu

Lesson 12

Local Extrema, the First and Second Derivative Tests

Initializations

```
> restart;
  with(oneonta):
>
```

12.1 The First Derivative Test

If c is a critical point of a continuous function f, then

- if f' changes from positive to negative at c, then f has a local maximum at c.
- if f' changes from negative to positive at c, then f has a local minimum at c.
- if f' does not change sign at c, then f has no local maximum or minimum at c.

The mathematical details will be provided in class.

Examples

Example 12.1.1

Compute the local maxima and minima of the function $f(x) = x^3(x-4)^4$.

Solution

Compute the critical points.

```
> f:=x->x^3*(x-4)^4;
```

$$f := x \to x^3 (x-4)^4 \quad \textbf{(2.1.1.1)}$$

```
> der1:=factor(diff(f(x), x));
```

$$der1 := x^2 (7x-12)(x-4)^3 \quad \textbf{(2.1.1.2)}$$

Since $f'(x)$ exists for all values of x, the zeros of $f'(x)$ are the only critical points of the function f. To avoid repetition of zeros we place the critical points in a Maple set. We do this by placing the **solve** command between braces.

```
> xc:={solve(der1=0, x)};
```

$$xc := \left\{0, 4, \frac{12}{7}\right\} \quad \textbf{(2.1.1.3)}$$

For convenience, we now put the critical points in a sorted Maple list.

```
> xc:=sort(convert(xc, list));
```

$$xc := \left[0, \frac{12}{7}, 4\right] \quad \textbf{(2.1.1.4)}$$

This has two advantages. First, the critical points are listed in increasing order, which make the information easier to read. Second, the elements in a list have a fixed place in the list, while the elements in a set do not have a fixed place.

Sketch $f(x)$ and the sign of its first derivative in one picture. The sign of $f'(x)$ is created using Maple's **signum** function defined by

$$\operatorname{signum}(x) = \begin{cases} -1 & \textbf{if} \quad x < 0 \\ 0 & \textbf{if} \quad x = 0 \\ 1 & \textbf{if} \quad x > 0 \end{cases}$$

Depending on the actual values of $f(x)$, you may want to multiply the sign of its derivative by an appropriate scalar.

```
> plot({f(x), 50*signum(der1)}, x=-1..5);
```



For the sake of convenience we recall the critical points.

```
> xc;
```

$$\left[0, \frac{12}{7}, 4\right] \qquad \textbf{(2.1.1.5)}$$

The function f has a local maximum at $\frac{12}{7}$ and a local minimum at 4. Since f' does not change sign at 0, the critical point 0 is not associated with an extremum. Maple makes quick work of finding the actual maximum and minimum values.

```
> max_min:=f(12/7), f(4);
```

$$max\_min := \frac{113246208}{823543}, 0 \qquad \textbf{(2.1.1.6)}$$

Maple's sequence **seq** command is a handy tool to generate a listing of the extrema in point format (x, y).

```
> extr:=seq([xc[k], f(xc[k])], k=2..3);
  extrf:=evalf(extr);
```

$$extr := \left[\frac{12}{7}, \frac{113246208}{823543}\right], [4, 0]$$

$$extrf := [1.714285714, 137.5109836], [4., 0.] \quad \textbf{(2.1.1.7)}$$

```
>
```

Example 12.1.2

Find the local extrema of $f(x) = x^{\frac{3}{5}}(4-x)^{\frac{1}{5}}$.

Solution

First, compute the critical numbers..

```
> f:=x->x^(3/5)*(4-x)^(1/5);
```

$$f := x \rightarrow x^{3/5} (4-x)^{1/5} \quad \textbf{(2.1.2.1)}$$

```
> der1:=simplify(diff(f(x), x));
```

$$der1 := -\frac{4}{5} \frac{x-3}{x^{2/5} (-x+4)^{4/5}} \quad \textbf{(2.1.2.2)}$$

The critical points are the zeros of the numerator and the denominator of this expression. That is 3, 0, and 4. We place them in increasing order in a Maple list.

```
> xc:=[0, 3, 4];
```

$$xc := [0, 3, 4] \quad \textbf{(2.1.2.3)}$$

If you wish to do so, you can let Maple generate the set of critical points by taking the union of the set of zeros of the numerator and the set of zeros of the denominator. Again, we place the results in an sorted Maple list.

```
> xc:={solve(numer(der1)=0, x)} union {solve(denom(der1)=0,
  x)};
  xc:=sort(convert(xc, list));
```

$$xc := \{0, 3, 4\}$$

$$xc := [0, 3, 4] \quad \textbf{(2.1.2.4)}$$

Make a plot to see where the extrema are located. Use the **realroots** routine in the **oneonta** package to make sure that the fractional powers evaluate to real values.

```
> plot(realroots({f(x), signum(der1)}), x=-2..6, numpoints=
  1000);
```



Observe that $f'(x)$ changes sign only at $x = 3$, so the critical points 0 and 4 are not associated with extrema. Clearly, $f(3)$ is a maximum for this function.

```
> extr:=[3, f(3)];
  evalf(extr);
```

$$extr := [3, 3^{3/5}]$$

$$[3., 1.933182045] \qquad (2.1.2.5)$$

```
>
```

12.2 The Second Derivative Test

Suppose f'' is continuous near c.

- If $f'(c) = 0$ and $f''(c) > 0$, then f has a local minimum at c.
- If $f'(c) = 0$ and $f''(c) < 0$, then f has a local maximum at c.

The advantage of the second derivative test is that it is easy to apply. A major disadvantage is that it only applies to critical points c where $f'(c) = 0$, and even then it will be inconclusive if $f''(c) = 0$.

Examples

Example 12.2.1

Let

$$\boldsymbol{f(x) = x^{\frac{2}{3}}(x-1)^3}$$

Do the following:
i) Use the second derivative test to find as many local extrema of *f* as you can.
ii) Use the first derivative test to find all the local extrema of *f*.

Solution
i) Use the second derivative test to find as many local extrema of *f* as you can.
First, plot the graph of this function.

```
> f:=x->x^(2/3)*(x-1)^3;
```

```
plot(realroots(f(x)), x=-1..2, y=-1..1);
```

$$f := x \to x^{2/3}\,(x-1)^3$$



From the picture it seems evident that *f* has at least one local maximum and one local minimum. We proceed by computing the critical points of *f*.

```
> der1:=factor(diff(f(x), x));
```

$$der1 := \frac{1}{3}\,\frac{(x-1)^2\,(11\,x-2)}{x^{1/3}} \qquad \textbf{(3.1.1.1)}$$

From the formula above we deduce that 0, $\frac{2}{11}$, and 1 are the three critical points of *f*. The second derivative test does not apply to 0, because $f'(0) \neq 0$. That leaves only two critical points to be tested, $\frac{2}{11}$ and 1. We compute $f''\left(\frac{2}{11}\right)$ and $f''(1)$.

```
> (D@@2)(f)(2/11);
  (D@@2)(f)(1);
```

$$\frac{27}{22}\,2^{2/3}\,11^{1/3}$$

$$0 \qquad \textbf{(3.1.1.2)}$$

Since $f''\left(\frac{2}{11}\right) > 0$, the function *f* has a local minimum at $\frac{2}{11}$. The second derivative test fails at the critical point 1. We compute $f\left(\frac{2}{11}\right)$.

```
> extr:=[2/11, f(2/11)];
  evalf(extr);
```

$$extr := \left[\frac{2}{11},\ -\frac{729}{14641}\,2^{2/3}\,11^{1/3}\right]$$

(3.1.1.3)

$$[0.1818181818, -0.1757819778] \quad (3.1.1.3)$$

In summary, the second derivative test allowed us to discover that the function *f* has a local minimum

$$f\left(\frac{2}{11}\right) = -\frac{729}{14641}\, 2^{2/3}\, 11^{1/3}$$

but it failed to give us any information about the critical points 0 and 1. All in all, a pretty dismal record for the second derivative test.

ii) Use the first derivative test to find all the local extrema of *f*.

Recall the first derivative of *f* and the critical points.

```
> der1;
```

$$\frac{1}{3}\,\frac{(x-1)^2\,(11\,x-2)}{x^{1/3}} \quad (3.1.1.4)$$

```
> xc:=[0, 2/11, 1];
```

$$xc := \left[0, \frac{2}{11}, 1\right] \quad (3.1.1.5)$$

Plot the function *f* and the sign of its first derivative in one picture.

```
> plot(realroots([f(x), 1/4*signum(der1)]), x=-1..2, y=-1..1, color=[red, blue]);
```



Clearly f' changes sign at 0 and at $\frac{2}{11}$, but not at 1. We conclude that the function *f* has a maximum $f(0)$ and a minimum $f\left(\frac{2}{11}\right)$, while the critical point 1 is not associated with an extremum.

```
> extr:=seq([xc[k],f(xc[k])], k=1..2);
  evalf(extr);
```

$$extr := [0, 0], \left[\frac{2}{11}, -\frac{729}{14641}\, 2^{2/3}\, 11^{1/3}\right]$$

$$[0., 0.], [0.1818181818, -0.1757819778] \quad \textbf{(3.1.1.6)}$$

>